Les séries géométriques

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue en septembre 2000 (Maple V). Ce document présente les séries géométriques dans le contexte d'un premier contact avec la notion même de série. Ce document permettera au lecteur de constater que parfois Maple donne des simplications inattendues. Loin d'être un inconvénient, cela donne l'occasion de se questionner sur les fondements mathématiques qui amènent Maple à donner de telles simplifications inattendues. C'est un enrichissement qui est un tantinet exigeant tout de même pour un lecteur faisant ses premiers pas avec Maple.

On retrouvera à la fin de la présentation des séries géométriques une petite procédure maison permettant d'illustrer une suite de polygones réguliers emboîtés d'un nombre *n* de côtés de son choix selon un décalage paramétrable par l'usager.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;
> with(plots,setoptions):
  setoptions(size=[300,300]):
> interface(imaginaryunit=i);
```

$$I \qquad (1.1)$$

L'initialisation suivante permettra d'avoir plus de lisibilité des nombres décimaux en supprimamt les zéros non significatifs à la fin d'un nombre.

```
> interface(typesetting=extended); # Pour s'assurer le niveau de
  composition étendue
  Typesetting:-Settings(striptrailing=true);
```

$$extended$$

$$false \qquad (1.2)$$

Suite géométrique

La suite $a_1, a_2, a_3, \ldots, a_{k-1}, a_k, a_{k+1}, \ldots$ est une suite géométrique si, et seulement si, par définition

– $a_1 \neq 0$,

– il existe un nombre réel $r \neq 0$ telle que $\dfrac{a_k}{a_{k-1}} = r$, pour tout entier $k > 1$.

Le nombre *r* est appelé la raison de la suite géométrique. Notons que la raison $r = \dfrac{a_k}{a_{k-1}}$ est le quotient de

toute paire de termes consécutifs de la suite.
Nous pouvons réécrire la formule précédente dans la forme équivalente $a_k = r\,a_{k-1}$, pour tout entier $k > 1$.

Ainsi, avec cette f*ormule de récurrence*, nous pouvons retrouver, à partir du premier terme, tous les termes de la suite en multipliant le terme précédent par la raison *r*.

Plus directement, le terme général de la suite géométrique est la formule $a_k = a_1\,r^{k-1}$, pour tout entier $k \geq 1$.
Cette formule suffit pour décrire tous les termes de la suite géométrique:

$$a_1, a_1\,r, a_1\,r^2, a_1\,r^3, a_1\,r^4, \ldots$$

On note ordinairement par *a* plutôt que par a_1 le premier terme d'une suite géométrique:

$$a, a\,r, a\,r^2, a\,r^3, a\,r^4, \ldots$$

Somme partielle des termes d'une suite géométrique:

$$\boldsymbol{S_n = \sum_{k=1}^{n} a\,r^{k-1}}$$

Une suite géométrique s'appelle aussi progression géométrique (PG en abrégé).

Crééons une fonction donnant le *k-ième* terme d'une progression géométrique. Nommons cette fonction PG.

```
> PG:=k->a*r^(k-1):
  a[k]=PG(k);
```

$$a_k = a\,r^{k-1} \tag{3.1}$$

La somme partielle des *n* premier termes d'une PG est une série géométrique finie. Cette somme partielle, $\sum_{k=1}^{n} a\,r^{k-1}$ est notée S_n.

```
> S[n]=Sum(PG(k),k=1..n);
```

$$S_n = \sum_{k=1}^{n} a\,r^{k-1} \tag{3.2}$$

Nous avons vu en classe que la somme partielle, S_n, des *n* premiers termes d'une suite géométrique est donnée par

$$S_n = \frac{a\left(1-r^n\right)}{1-r}, \quad r \neq 1$$

Interrogeons Maple pour voir s'il reconnaît formellement cette somme partielle.

```
> Sum(PG(k),k=1..n)=sum(PG(k),k=1..n);
  Formule_A:=S[n]=simplify(rhs(%));
```

$$\sum_{k=1}^{n} a\,r^{k-1} = \frac{a\,r^{n+1}}{r\,(r-1)} - \frac{a}{r-1} \tag{3.3}$$

$$Formule\_A := S_n = \frac{a\left(r^n - 1\right)}{r - 1} \quad (3.3)$$

Puisque $a_n = a\,r^{n-1}$ ($r \neq 0$), la formule devient $S_n = \dfrac{r\,a_n - a}{r - 1}$, $r \neq 1$

Obtenons cette formule avec Maple.

```
> subs(r^n=a[n]*r/a, Formule_A);
```

$$S_n = \frac{a\left(\frac{a_n r}{a} - 1\right)}{r - 1} \quad (3.4)$$

```
> simplify(rhs((3.4)));
```

$$\frac{a_n r - a}{r - 1} \quad (3.5)$$

```
> Formule_B:=S[n]=normal((3.5),expanded);
```

$$Formule\_B := S_n = \frac{a_n r - a}{r - 1} \quad (3.6)$$

Voilà donc deux formules pour obtenir la somme partielle S_n si, dans chacun des cas, $r \neq 1$.

```
> Formule_A;
  Formule_B;
```

$$S_n = \frac{a\left(r^n - 1\right)}{r - 1}$$

$$S_n = \frac{a_n r - a}{r - 1} \quad (3.7)$$

– Formule_A lorsque le terme général de la suite est donné

– Formule_B lorsque le premier terme, le dernier terme et la raison de la suite sont donnés.

Exemples d'application des formules $S_n = \dfrac{a\left(r^n - 1\right)}{r - 1}$ et $S_n = \dfrac{a_n r - a}{r - 1}$

Exemple 1

Obtenons la somme partielle des 15 premières valeurs de la suite géométrique

$\dfrac{3}{14}, \dfrac{9}{28}, \dfrac{27}{56}, \dfrac{81}{112}, \dfrac{243}{224}, \dfrac{729}{448}, ...$

Trouvons la raison en divisant le deuxième terme par le premier.

```
> r=(9/28)/(3/14);
```

$$r = \frac{3}{2} \tag{4.1.1}$$

Vérification des premiers termes de la progression géométrique: le premier terme est évidemment $\frac{3}{14}$, alors le terme général est

```
> PG:=k->3/14*(3/2)^(k-1);
  seq(PG(k),k=1..6);
```

$$PG := k \mapsto \frac{3 \cdot \left(\frac{3}{2}\right)^{k-1}}{14}$$

$$\frac{3}{14}, \frac{9}{28}, \frac{27}{56}, \frac{81}{112}, \frac{243}{224}, \frac{729}{448} \tag{4.1.2}$$

Le calcul direct avec la macro-commande `add` nous donne

```
> Sum(PG(k),k=1..15)=add(PG(k),k=1..15);
```

$$\sum_{k=1}^{15} \frac{3\left(\frac{3}{2}\right)^{k-1}}{14} = \frac{42948417}{229376} \tag{4.1.3}$$

Vérifions avec la première formule.

```
> eval(Formule_A,[a=3/14,r=3/2,n=15]);
```

$$S_{15} = \frac{42948417}{229376} \tag{4.1.4}$$

Vérifions avec la seconde formule.

```
> eval(Formule_B,[r=3/2,a=3/14,n=15,a[n]=PG(15)]);
```

$$S_{15} = \frac{42948417}{229376} \tag{4.1.5}$$

Exemple 2

Calculons le nombre de termes n et la raison r de la PG dont le premier terme est 256, le dernier terme 81 et la somme des n termes 781.

Solution:

Utilisons la formule Formule_B: $S_n = \frac{a_n r - a}{r-1}$. Résolvons, pour r, l'équation $781 = \frac{81\, r - 256}{r-1}$

```
> subs(S[n]=781,a[n]=81,a=256,Formule_B);
  r=solve(%,r);
```

$$781 = \frac{-256 + 81\, r}{r-1}$$

$$r = \frac{3}{4} \tag{4.2.1}$$

Puisque $a_k = a\, r^{k-1}$, il suffit de résoudre pour k l'équation $81 = 256 \left(\frac{3}{4}\right)^{k-1}$

```
> Équation:=81=256*(3/4)^(k-1);
  n=solve(Équation,k);
```

$$Équation := 81 = 256\left(\frac{3}{4}\right)^{k-1}$$

$$n = \frac{\ln\left(\frac{3}{4}\right) + \ln\left(\frac{81}{256}\right)}{\ln\left(\frac{3}{4}\right)} \tag{4.2.2}$$

```
> simplify((4.2.2));
```

$$n = 5 \tag{4.2.3}$$

Somme de tous les termes d'une suite géométrique : $\sum_{k=1}^{\infty} a\,r^{k-1}$

La somme de tous les termes d'une progression géométrique, $a + a\,r + a\,r^2 + a\,r^3 + \;... +\; a\,r^{n-1} + ...$ se note $\sum_{k=1}^{\infty} a\,r^{k-1}$.

L'expression $\sum_{k=1}^{\infty} a\,r^{k-1}$ est appelée **série géométrique**.

Rappelons la formule de la somme partielle des n premiers termes d'une série géométrique $S_n = \frac{a\left(1 - r^n\right)}{1 - r}$ peut s'écrire, avant simplification

$$S_n = \frac{a}{1-r} - \frac{a\,r^n}{1-r} \quad \text{lorsque bien sûr } r \neq 1.$$

Si $-1 < r < 1$, i.e. si $|r| < 1$, la différence entre S_n et $\frac{a}{1-r}$ devient aussi petite que nous le désirerons pourvu que nous prenions n suffisamment grand.

Dans ce cas, nous écrivons $\lim_{n\to\infty} S_n = \lim_{n\to\infty}\left(\frac{a}{1-r} - \frac{a\,r^n}{1-r}\right) = \frac{a}{1-r}$ car $\lim_{n\to\infty}\frac{a\,r^n}{1-r} = 0$.

En notant cette limite S et nous écrivons $S = \frac{a}{1-r}$, $|\,r\,| < 1$.

Le mécanisme de la simplification de Maple connaît très bien la formule de la somme de tous les termes d'une suite géométrique.

```
> PG:=k->a*r^(k-1);
```

$$PG := k \mapsto a \cdot r^{k-1} \tag{5.1}$$

```
> Sum(PG(k),k=1..infinity)=sum(PG(k),k=1..infinity,formal=true);
```

$$\sum_{k=1}^{\infty} a\,r^{k-1} = -\frac{a}{r-1} \tag{5.2}$$

```
> ``=subs([a=-a,r-1=1-r],rhs((5.2)));
```

$$= \frac{a}{1-r} \qquad \textbf{(5.3)}$$

Exemples d'application de la formule $S = \frac{a}{1-r}$

ATTENTION: Cette simplification automatique n'est valable seulement que dans le domaine de simplification de l'expression, c'est-à-dire pour $|r| < 1$. Maple a identifié formellement que le terme général de la série était celui d'une série géométrique. En spécifiant l'option *formal=true*, Maple a donc simplifié automatiquement

$$S = \sum_{k=1}^{\infty} a\, r^{k-1} \quad \text{par la formule } \frac{a}{1-r}.$$

Notons que la manière d'envisager l'addition d'un nombre infini de termes correspond à la limite de la somme partielle des n premiers termes lorsque n croît indéfiniment. En effet, la notation mathématique $\sum_{k=1}^{\infty} a\, r^{k-1}$ est un abus de langage. En analyse mathématique, cette notation est une écriture abrégée du calcul de la limite suivante:

$$\sum_{k=1}^{\infty} a\, r^{k-1} = \lim_{n \to \infty} \left(\sum_{k=1}^{n} a\, r^{k-1} \right).$$

Maple n'a pas pu évaluer, évidemment, cette limite qui est, comme nous l'avons vu en classe, une évaluation qui dépend de la valeur de la raison r. Donc, la formule $-\frac{a}{r-1}$ est une simplification *formelle* de $\sum_{k=1}^{\infty} a\, r^{k-1}$ où il est sous-entendu que $|r| < 1$.

Exemple 1

Calculons la "somme" de la progression géométrique $3 + \frac{3}{2} + \frac{3}{4} + \ldots$ de deux manières:

– en notation abrégée $\sum_{k=1}^{\infty} a\, r^{k-1}$ et

– par le calcul de la limite $\lim_{n \to \infty} \left(\sum_{k=1}^{n} a\, r^{k-1} \right)$.

C'est une PG où $a = 3$ et $r = \frac{1}{2}$.

Pour plus de commodité dans les développements ci-dessous, reformulons la fonction à une variable PG donnant le k-ième terme d'une progression géométrique par une fonction de deux variables donnant le terme général.

```
> PG:=(a,r)->a*r^(k-1);
```

$$PG := (a, r) \mapsto a \cdot r^{k-1} \qquad \textbf{(6.1.1)}$$

En notation abrégée utilisons la macro-commande `sum` en explicitant directement l'infini comme borne supérieure :

```
> Sum(PG(3,1/2),k=1..infinity)=sum(PG(3,1/2),k=1..infinity);
```

$$\sum_{k=1}^{\infty} 3\left(\frac{1}{2}\right)^{k-1} = 6 \qquad \textbf{(6.1.2)}$$

Par le calcul de la limite :

```
> Limit(Sum(PG(3,1/2),k = 1 .. n),n = infinity)=Limit(sum(PG(3,1/2),
  k = 1 .. n),n = infinity);
  ``=limit(sum(PG(3,1/2),k = 1 .. n),n = infinity);
```

$$\lim_{n\to\infty} \sum_{k=1}^{n} 3\left(\frac{1}{2}\right)^{k-1} = \lim_{n\to\infty}\left(-12\left(\frac{1}{2}\right)^{n+1} + 6\right)$$

$$= 6 \qquad \textbf{(6.1.3)}$$

Exemple 2

Déterminons le nombre rationnel égale au nombre décimal périodique 0,36363636...

Ce nombre s'écrit dans une forme équivalente

0,36 + 0,0036 + 0,000036 + ...

Considérons une progression géométrique dont le premier terme est $a = \frac{36}{100}$ et la raison $r = \frac{1}{100}$. D'où

```
> PG:=(a,r)->a*r^(k-1);
```

$$PG := (a, r) \mapsto a \cdot r^{k-1} \qquad \textbf{(6.2.1)}$$

```
> Sum(PG(36/100,(1/100)),k=1..infinity)=sum(PG(36/100,(1/100)),k=1..
  infinity);
```

$$\sum_{k=1}^{\infty} \frac{9\left(\frac{1}{100}\right)^{k-1}}{25} = \frac{4}{11} \qquad \textbf{(6.2.2)}$$

Exemple 3

Déterminons le nombre rationnel égal au nombre décimal périodique 2,35242424...

2,35 + 0,0024 + 0,000024 + 0,00000024 + ...

Considérons la somme de 2,35 et d'une série géométrique dont le premier terme est 0,0024 et la raison 0, 01. D'où

```
> PG:=(a,r)->a*r^(k-1);
```

$$PG := (a, r) \mapsto a \cdot r^{k-1} \qquad \textbf{(6.3.1)}$$

```
> Nombre=2.35+Sum(PG(0.0024,(0.01)),k=1..infinity);
```

$$Nombre = 2.35 + \left(\sum_{k=1}^{\infty} 0.0024\, 0.01^{k-1}\right) \qquad \textbf{(6.3.2)}$$

```
> ``=value(rhs((6.3.2)));
```

(6.3.3)

$$= 2.352424242 \tag{6.3.3}$$

```
> convert((6.3.3),rational);
```

$$= \frac{7763}{3300} \tag{6.3.4}$$

À propos de la simplification numérique de $\sum_{k=1}^{\infty} a\, r^{k-1}$ avec $r \geq 1$ ou $r \leq -1$

On sait que la somme S n'existe pas lorsque $r \geq |1|$, c'est-à-dire lorsque $r \geq 1$ ou lorsque $r \leq -1$. Explorons le calcul de telles sommes effectuées par Maple.

Reformulons à nouveau une fonction PG donnant le *k-ième* terme d'une progression géométrique en fonction du premier terme a et de la raison r.

```
> PG:=(a,r)->a*r^(k-1);
```

$$PG := (a, r) \mapsto a \cdot r^{k-1} \tag{7.1}$$

Obtenons avec Maple la somme d'une série géométrique dont le premier terme est $a = 1$ et dont la raison $r = \frac{3}{2}$.

```
> Sum(PG(1,3/2),k=1..infinity)=sum(PG(1,3/2),k=1..infinity);
```

$$\sum_{k=1}^{\infty} \left(\frac{3}{2}\right)^{k-1} = \infty \tag{7.2}$$

Voilà ! Un résultat tout à fait attendu. Observons le résultat avec une raison $r = 1$.

```
> Sum(PG(1,1),k=1..infinity)=sum(PG(1,1),k=1..infinity);
```

$$\sum_{k=1}^{\infty} 1 = \infty \tag{7.3}$$

Encore là, un résultat conforme à la théorie. Aucune surprise ! Voyons avec une raison $r = -1$.

```
> Sum(PG(1,-1),k=1..infinity)=sum(PG(1,-1),k=1..infinity);
```

$$\sum_{k=1}^{\infty} (-1)^{k-1} = \sum_{k=1}^{\infty} (-1)^{k-1} \tag{7.4}$$

Dans le membre de droite, la mise en forme de la sommation en couleur bleu nous indique que le mécanisme de la simplification automatique de Maple n'a pu donner un résultat "clair". Curieux n'est-ce pas ? La théorie vue en classe nous a pourtant montré clairement que la série diverge dans ce cas et cela devrait être à la portée de Maple de faire cette simplification. Un petit complément d'information est nécessaire pour expliquer un tel résultat.

En général, dans le cas des séries divergentes et en particulier pour les séries géométriques divergentes, le mécanisme de la simplification automatique produira l'un des résultats suivants: ∞, $-\infty$, ou l'expression sérielle elle-même. Lorsque le résultat est, comme un écho, soit la série elle-même, c'est probablement parce qu'il est possible d'*avoir une valeur numérique pour la série divergente.* Pour certaines classes de séries divergentes, on peut leur appliquer des méthodes de sommation dite de Cesaro ou d'Abel ou encore leur accorder une valeur comme c'est le cas pour réaliser un prolongement analytique de continuité pour la fonction zêta ζ de Riemann. Il serait tout à fait indiqué d'avoir, en option de la macro-commande `sum`, la possibilité d'effectuer une sommation explicitement dans le sens de Cesaro ou d'Abel par exemples. En ne précisant aucune de ces options, nous pourrions avoir un résultat clair pour une série divergente au sens habituel, c'est-à-dire comme la limite à l'infini de la suite des sommes partielles. Malheureusement, ce n'est pas le cas. Par contre, il est possible par l'initialisation de la variable d'environnement «_EnvFormal» de privilégier les sommations divergentes dans le sens de Cesaro et al.

Ouf, ces explications sont loin d'être claires pour des élèves faisant leur premier pas dans l'univers des séries et, en plus, avec comme seul modèle le modèle des séries géométriques. Ce qui est à retenir somme toute, c'est que lorsque le résultat d'une requête de sommation est la sommation elle-même, il faudra s'assurer autrement de la divergence (au sens habituel) de la série étudiée.

Considérons à nouveau la série géométrique $\sum_{k=1}^{\infty} (-1)^{k-1}$ qui est appelée série d'Euler. Bien que nous sachions que cette série est divergente au sens habituel, appliquons-lui tout de même une requête d'évaluation numérique avec la macro-commande `evalf`.

```
> Sum(PG(1,-1),k=1..infinity)=sum(PG(1,-1),k=1..infinity);
```

$$\sum_{k=1}^{\infty} (-1)^{k-1} = \sum_{k=1}^{\infty} (-1)^{k-1} \tag{7.5}$$

```
> ``=evalf(rhs((7.5)));
```

$$= 0.5 \tag{7.6}$$

La valeur 0,5 découlant de l'évaluation numérique montre effectivement que la sommation n'a pas été faite au sens habituel et, même si, au départ, Maple n'a donné aucune indication, il ne s'agit pas du tout d'une erreur.

Pour montrer la divergence de cette série géométrique, traitons explicitement cette sommation au sens habituel en évaluant la limite à l'infini de la suite des sommes partielles. C'est possible car nous pouvons obtenir facilement l'expression du terme général de la suite des sommes partielles, expression qui correspond à la formule donnant la somme des *n* premiers termes d'une PG.

```
> Limit(Sum(PG(1,-1),k=1..n),n=infinity)=Limit(sum(PG(1,-1),k=1..n),n=
  infinity);
```

$$\lim_{n \to \infty} \sum_{k=1}^{n} (-1)^{k-1} = \lim_{n \to \infty} \left(\frac{(-1)^{n+1}}{2} + \frac{1}{2} \right) \tag{7.7}$$

```
> assume(n,even);
  Limit(Sum(PG(1,-1),k=1..n),n=infinity)=Limit(sum(PG(1,-1),k=1..n),n=
  infinity);
  ``=value(rhs(%));
```

$$\lim_{n \to \infty} \sum_{k=1}^{n} (-1)^{k-1} = \lim_{n \to \infty} 0$$

$$= 0 \tag{7.8}$$

```
> assume(n,odd);
  Limit(Sum(PG(1,-1),k=1..n),n=infinity)=Limit(sum(PG(1,-1),k=1..n),n=
```

```
infinity);
``=value(rhs(%));
```

$$\lim_{n \to \infty} \sum_{k=1}^{n} (-1)^{k-1} = \lim_{n \to \infty} 1$$

$$= 1 \qquad (7.9)$$

Ce résultat précédent montre que la limite à l'infini de la suite des sommes partielles n'existe pas et donc que la série géométrique diverge (au sens habituel).

```
> unassign('n');
```

Une autre manière de montrer que la série diverge est d'établir une liste des sommes partielles successives des n premiers termes.

La petite procédure suivante servira à bien présenter les calculs successifs.

```
> Tableau:=proc(Expr,r1::name=range)
   local Début,Fin,j,k,n;
   k:=lhs(r1);
   Début:=lhs(rhs(r1));
   Fin:=rhs(rhs(r1));
   matrix([[` n`,`||`,Sum(Expr,k=Début..n)],
          [`-------`,`||`,`-----------------`],
          seq([ n,`||`,sum(Expr,k=Début..n)],n=Début..Fin)]);
  end proc:
```

Obtenons maintenant le tableau des 10 premières sommes partielles de la série $\sum_{k=1}^{\infty} (-1)^{k-1}$

```
> Tableau(PG(1,-1),k=1..10);
```

$$\left[\begin{array}{ccc} n & || & \sum_{k=1}^{n} (-1)^{k-1} \\ ------- & || & ----------------- \\ 1 & || & 1 \\ 2 & || & 0 \\ 3 & || & 1 \\ 4 & || & 0 \\ 5 & || & 1 \\ 6 & || & 0 \\ 7 & || & 1 \\ 8 & || & 0 \\ 9 & || & 1 \\ 10 & || & 0 \end{array}\right] \qquad (7.10)$$

Le tableau montre bien que les sommes partielles oscillent constamment entre la valeur 1 et et la valeur 0.

```
> limit(PG(1,-1),k=infinity);
```

$$-1 - \mathrm{i}..1 + \mathrm{i} \tag{7.11}$$

Considérons maintenant une autre série géométrique. Soit la série géométrique de raison $r = -\frac{3}{2}$ (de premier terme $a = 1$). Nous savons que cette série géométrique est **divergente**.

```
> Sum(PG(1,-3/2),k=1..infinity)=sum(PG(1,-3/2),k=1..infinity);
  ``=evalf(rhs(%));
```

$$\sum_{k=1}^{\infty} \left(-\frac{3}{2}\right)^{k-1} = \sum_{k=1}^{\infty} \left(-\frac{3}{2}\right)^{k-1}$$

$$= 0.4 \tag{7.12}$$

Traitons cette série au sens habituel.

```
> Limit(Sum(PG(1,-3/2),k=1..n),n=infinity)=Limit(sum(PG(1,-3/2),k=1..
  n),n=infinity);
```

$$\lim_{n\to\infty} \sum_{k=1}^{n} \left(-\frac{3}{2}\right)^{k-1} = \lim_{n\to\infty} \left(\frac{4\left(-\frac{3}{2}\right)^{n+1}}{15} + \frac{2}{5}\right) \tag{7.13}$$

Comme il s'agit de la puissance entière d'un nombre négatif, le calcul de la limite dépend de la parité de *n*.

```
> assume(n,even);
  Limit(sum(PG(1,-3/2),k=1..n),n=infinity)=limit(sum(PG(1,-3/2),k=1..
  n),n=infinity);
```

$$\lim_{n\to\infty} \left(-\frac{2^{1-n}3^{n}}{5} + \frac{2}{5}\right) = -\infty \tag{7.14}$$

```
> unassign('n');
```

Puisque cette limite n'existe pas, cela est suffisant pour montrer la divergence de la série géométrique $\sum_{k=1}^{\infty} \left(-\frac{3}{2}\right)^{k-1}$.

Ce n'est pas seulement dans le cas d'une série géométrique divergente où nous pouvons avoir une divergence dans le sens habituel et d'avoir, en même temps, la possibilité que soit donné une valeur numérique à la série divergente. C'est le cas avec les séries de Riemann divergentes.

D'une part, nous avons

```
> Sum(1/k^(1/3),k=1..infinity)=sum(1/k^(1/3),k=1..infinity);
```

$$\sum_{k=1}^{\infty} \frac{1}{k^{1/3}} = \infty \tag{7.15}$$

D'autre part, nous avons

```
> Sum(1/k^(1/3),k=1..infinity)=evalf(Sum(1/k^(1/3),k=1..infinity));
```

$$\sum_{k=1}^{\infty} \frac{1}{k^{1/3}} = -0.9733602484 \tag{7.16}$$

Résultat numérique encore plus bizarre que dans les cas des séries géométriques précédentes, nous avons comme résultat numérique à cette série divergente une valeur négative bien qu'il s'agisse d'additionner un *nombre infini de termes positifs...*

Gardons toujours à l'esprit que Maple est un logiciel de calcul formel puissant qui a pour objet de transposer une grande partie des mathématiques à l'ordinateur. Le domaine des mathématiques étant très vaste, la mise au point du mécanisme de la simplification automatique peut très bien tenir compte de notions mathématiques pouvant être inconnues à notre niveau d'enseignement. C'est le cas de la sommation au sens de Cesaro qui permet d'assigner une valeur numérique à une série divergente (séries géométriques divergentes et autres). Cette notion n'est abordée dans des cours d'analyse avancées. Cela a donné, par contre, une excellente occasion de rechercher une explication à de tels résultats "mystifiants" de prime abord. Il ne faut donc pas conclure trop rapidement à un "bug" du logiciel.

Pour d'autres résultats, leur incongruité peut découler tout simplement d'erreurs dans le mécanisme de la simplification automatique ou d'erreurs de programmation et ça, c'est l'affaire des mathématiciens et des mathématiciennes de les découvrir et de les rapporter. Dans ce cas, nous pouvons effectivement parler de "bug" du logiciel. Mais, heureusement, cela se produit rarement. De plus, il faut aussi réaliser que méconnaître la limite physique de notre machine peut nous amener à obtenir des résultats bizarres. Finalement, l'usager lui-même peut être une source d'erreur non négligeable...

En résumé, dans le cas du doute d'un éventuel résultat obtenu avec Maple, l'usager doit explorer toutes ces possibilités à la lumière de ses acquis mathématiques. En clair, dans le cas d'un doute, l'usager doit s'assurer autrement (à l'aide du logiciel lui-même ou autrement) de la validité du résultat obtenu car *c'est l'usager qui est le mathématicien et non la machine.*

Procédure des polygones emboîtés

Voici une procédure amusante servant à illustrer la construction d'une suite de polygones réguliers emboîtés tels que les sommets de chaque polygone sont sur les arêtes du polygone précédent décalés d'une fraction de la longueur de l'arête du polygone précédent.

Le lien de tels constructions avec les séries géométriques est fait dans la partie des exercices de ce document.

Macro-commande Poly_Emboîtés

Poly_Emboîtés(Nombre_de_côtés, Nombres_d'emboîtements, Décalage, Affichage)

La macro-commmande Poly_Emboîtes construit une séquence de polygones réguliers emboîtés. Le polygone P_{k+1} est construit à partir du polygone P_k. Chacun des sommets du polygone P_{k+1} sont sur les arêtes du polygone P_k situés à une fraction de la distance de chaque sommet du polygone P_k.

- **Nombres_de_côtés** est un nombre entier supérieur à 2. C'est le nombre de côtés du polygone régulier initial
- **Nombres_d'emboîtements** est un nombre entier supérieur ou égal à 0. C'est le nombre de polygones semblables inscrits
- **Décalage** est un nombre réel de l'intervalle [0, 1].
- **Affichage** est le type d'affichage des polygones: fixe ou animé.

Initialisation

Positionner le curseur clignotant sur une ligne quelconque du bloc de requêtes ci-dessous et pressez la touche Entrée. Vous pourrez ensuite cliquer sur le bouton (le carré gris avec le signe moins) de cette sous-section pour la rétracter.

```
> Poly_Emboîtés:=proc(Nbc,Nbe,Décal,Mode)
    #-----------------------------------------
    # Création des variables locales
    local Centre,Décalage,j,k,Nbr_Côtés,Nbr_Emboîtements,Mod,Options,
  Points,S,X0,
    X1,Y0,Y1:
    #-----------------------------------------
    # Validation des arguments de la procédure
    if evalf(Nbc)<3 or not type(Nbc,posint) then
     ERROR("Vous devez donner au moins un nombre entier de côté égale à
  trois")
    fi;
    if evalf(Nbe)<0 or not type(Nbc,posint) then
     ERROR("Vous devez donner au moins un nombre entier d'emboîtements
  égale à 0")
    fi;
    if evalf(Décal)<0 or evalf(Décal)>1 then
    ERROR("Vous devez donner une valeur de décalage comprise entre 0 et 1
  inclusivement")
    fi;
    if not member(Mode,{animé,fixe})  then
      ERROR("Vous devez spécifier soit animé ou soit fixe")
    fi;
    #-----------------------------------------
    # Initialisation des variables
    Nbr_Côtés:=Nbc:
    Nbr_Emboîtements:=Nbe+1:
    Décalage:=evalf(Décal):
    Mod:=Mode:
    Centre:=[0,0]:
    S:=evalf(2*Pi/Nbr_Côtés):
    if is(Nbr_Côtés,odd) then X0:=sin(S): Y0:=cos(S)
     else X0:=cos(Pi/2)-sin(S): Y0:=sin(Pi/2)+cos(S)
    fi;
    #-----------------------------------------
    # Création des images
    for j from 1 to Nbr_Emboîtements do
     Points:=[X0,Y0]:
     for k from Nbr_Côtés by -1 to 1 do
      X1:=Centre[1]+(X0-Centre[1])*cos(S)+(Y0-Centre[2])*sin(S):
      Y1:=Centre[2]-(X0-Centre[1])*sin(S)+(Y0-Centre[2])*cos(S):
      if k<>1 then X0:=X1: Y0:=Y1 fi:
      Points:=Points,[X1,Y1]:
     od:
     if type(j,even) then Options:=color=turquoise else
      Options:=color=pink
     fi:
    Figure||j:=plottools[polygon]([Points],Options,linestyle=0):
    Polygone||(Nbr_Emboîtements+1-j):=plots[display](Figure||j):
    Polyanime||j:=plots[display]([Polygone||((Nbr_Emboîtements+1-j)..
  Nbr_Emboîtements)]):
    X0:=X0+Décalage*(X1-X0):
    Y0:=Y0+Décalage*(Y1-Y0):
    od:
    #------------------------------------------
    # Superposition des images
    if Mod=fixe then plots[display]([Polygone||(1..Nbr_Emboîtements)],
       scaling=constrained,axes=none)
```

```
        elif Mod=animé then plots[display]([Polyanime||(1..Nbr_Emboîtements)
],
        insequence=true,
        scaling=constrained,axes=none)
        fi
    end:
```

Exemples

```
> Poly_Emboîtés(3,24,sqrt(3)/2,bof);
```

Error, (in Poly_Emboîtés) Vous devez spécifier soit animé ou soit fixe

```
> Poly_Emboîtés(5,48,sqrt(3)/2,animé);
```



```
> Poly_Emboîtés(4,36,0.25,animé);
```



```
> Poly_Emboîtés(4,36,0,fixe);
```



```
> Poly_Emboîtés(5,40,.15,fixe);
```



```
> Poly_Emboîtés(7,50,0.5,animé);
```



Exercices

No.1

Calculer la somme des séries géométriques convergentes suivantes.

a) $12 + 6 + 3 + ...$

b) $5 + 1 + \frac{1}{5} + ...$

c) $0,9 + 0,03 + 0,001 + ...$

d) $60 + 6 + 0,6 + ...$

e) $22 - 2 + \frac{2}{11} - ...$

f) $1 - \frac{1}{4} + \frac{1}{16} - \frac{1}{64} + ...$

No.2

Exprimer les nombres décimaux périodiques suivants sous leur écriture rationnelle
- en exprimant le nombre décimal comme une progression géométrique
- à l'aide de la macro-commande `convert( , rational)`

Vérifier votre réponse en effectuant la division.

a) 0,777777...
b) 1,32454545...
c) 0,124124124124...

d) 0,69696969...
e) 7,1272727...
f) 27,279279279...
g) 0,64705882352971171647058823529711716 4...

No.3

Soit 9/4 la somme d'une progression géométrique dont le premier terme est 3/2. Quelle en est la raison ?

No.4

L'évaluation par Maple de la série $\sum_{k=2}^{\infty} \frac{(-1)^k 2^{k+1}}{3^{k-3}}$ est $\frac{72}{5}$. Cette réponse est-elle exacte ? Justifier.

No.5

Soit la suite de carrés emboîtés C_1, C_2, C_3,...,C_k,... illustrée ci-dessous.



Le décalage de l'emboîtement est la fraction $\frac{\sqrt{3}}{6}\ a_k$

Soit a_k, A_k et P_k respectivement le côté, l'aire et le périmètre du carré C_k.

a) Calculer la relation entre a_{k+1} et a_k.
b) Calculer a_n, A_n et P_n.
c) Calculer $\sum_{n=1}^{\infty} P_n$.

Développer tous vos calculs symboliquement et non pas avec des décimales.

No.6

Soit la série infinie de carrés emboîtés avec un décalage de 0,5. Les trois premières figures vous donne la manière de colorer en bleu une région de chaque carré emboîté.



Calculer l'aire totale de la région en bleu si la mesure de l'arête du premier carré vaut deux unités.

No.7

Pour les champions, même question que le numéro 5, même décalage mais avec un pentagone régulier.
Suggestion: utilisez la loi des cosinus.